

Paper 048-2007

Categorizing Drug Data with SAS® PROC FORMAT, INPUT and PUT Functions

Henry Xia, CDC/National Center for Health Statistics

ABSTRACT

Drug data recoding is commonly used in analyses of drug utilization rates, drug risks and side effects, and drug abuse and its prevention. Classification of a specific drug into a predetermined category using drug codes is complex. There are more than 100,000 brands of drugs on the market in the United States with more than 40,000 generic name codes, and some drugs are combinations of different ingredients. In addition, some drugs have multiple therapeutic uses, and thus, the most appropriate drug category may vary depending on the specific analysis. In this paper, a flexible SAS macro is presented for recoding drug data into groups using the SAS INPUT() and PUT() functions according to a format defined for drug codes in the FORMAT procedure. The macro can be used for any listed drugs in a data set. The macro can categorize each drug into up to ten categories without code modification, which greatly simplifies the coding process. The macro also can be used for categorizing other data into meaningful groups or classification systems, such as ICD (International Classification of Diseases) diagnosis codes or CPT (Current Procedural Terminology) procedure codes. An example of recoding individual psychotropic drugs (recorded either as ingredients, generic drug name, or proprietary drug name) into major psychotropic drug categories is presented using data from the 2002 National Ambulatory Medical Care Survey (NAMCS). The TABULATE procedure is used to produce statistical reports about the percentage of visits with a prescribed psychotropic drug. ODS (Output Delivery System) is used to deliver the report into a table in HTML format.

INTRODUCTION

Drugs are frequently used therapies for reducing morbidity and mortality, and improving the quality of life of Americans. National Health and Nutrition Examination Survey (NHANES) data for 1999-2000 show that 44.3 percent of Americans of all ages reported using any prescribed medication during the past month. There are more than 100,000 brands of drugs on the market in the United States with more than 40,000 generic name codes, and some drugs are combinations of different ingredients. Drug data recoding is commonly used in research to produce meaningful therapeutic groups that can be used in analyses of drug utilization rates, drug risks and side effects, and drug abuse and its prevention. The complicated drug recoding process can be simplified by using the SAS INPUT() and PUT() functions and macro language to convert drug codes into simple categorical values according to a FORMAT defined for drug codes. An example of recoding individual psychotropic drugs (recorded either as ingredients, generic drug names, or proprietary drug names) into major psychotropic drug categories is presented based on data from the 2002 National Ambulatory Medical Care Survey (NAMCS). The TABULATE procedure is used to produce a tabular report with weighted and unweighted percentages of visits by psychotropic drug category and gender. ODS (Output Delivery System) is used to deliver the report on psychotropic drugs from PROC TABULATE into a table in HTML format. Although the primary audience of the paper is programmers or others who do the actual recoding, the paper might also be of interest to data analysts seeking to understand how these variables were recoded so that they can apply the methodology/programming code to data requiring a similar type of categorization.

OBJECTIVES

- Illustrate a flexible SAS macro to simplify the drug data recoding process using the SAS INPUT() and PUT() functions and PROC FORMAT.
- Present a statistical report on the categorized drug data using PROC TABULATE and ODS.

DATA

The National Ambulatory Medical Care Survey (NAMCS) is a nationally representative probability sample survey of visits to office-based physicians that is conducted by the National Center for Health Statistics (NCHS), Centers for Disease Control and Prevention (CDC). Up to six medications were recorded during each visit in the 2002 NAMCS. Medications that are recorded include prescriptions made during the visit, medications that the physician recommended the patient continue taking, and medications actually provided to the patient during the visit. Drug-related information available for each drug recorded in the 2002 NAMCS is shown in Table 1. The drug entry code is from a unique classification scheme developed at NCHS. The generic name code is a unique code assigned to each official generic drug name or entity by the United States Pharmacopeia or other responsible authority. The composition status code indicates whether a medication is a single-ingredient or a combination drug. A combination drug can be identified by up to 5 ingredients. The ingredient codes are used to identify the primary class to which each of up to 5 drug ingredients may belong. The NDC therapeutic class is a 4-digit code used to identify the primary therapeutic class for which the drug is used, based on the National Drug Code Directory therapeutic classification system.

Table 1. Drug-related information in the 2002 NAMCS data file

Drug entry code	Generic name code	Prescription status code	Composition status code	National Drug Code (NDC) Therapeutic Class	Ingredient1 – Ingredient5
00001-99227 99980- undetermined 99999- illegible 90000- blank	50001-51379 51383-92512 50000 – undetermined 51380-51382 - combination product	1-prescribed drug 2-non prescribed drug 3-undetermined	1-Single entity 2-Combination drug 3-undetermined	0100-2100	50001-51379 51383-92512 50000 – undetermined 51380-51382 - combination product
Variable names that are used in the SAS data set					
MEDICAT#	GEN#	PRESCRI#	COMSTATS#	NDCCLAS#	DRG#ING1-DRG#ING5
# = 1, 2, 3, 4, 5, 6 denotes medication number from 1 to 6					

Two technical terms used in the paper are defined as follows:

- First-listed drug – This is the medication or drug that is recorded first out of all drugs recorded (up to the maximum possible number of fields in a given data year— up to six drugs can be listed in the 2002 NAMCS data, and up to eight drugs can be listed in the 2003 NAMCS and later years). Because the number of visits with one drug is much larger than the number of visits with two, three, four, five, and six drugs recorded during a visit, the first-listed drug is often considered the most salient or important drug recorded for a particular visit. Some analysts may use only the first-listed drug in their analysis because the process of recoding all-listed drugs is complicated.
- Any-listed drug -- This is defined as any single drug from all of the drugs recorded during a visit, and each visit is counted only once. For example, it could be the first, the fourth or the eighth drug in the order of drugs listed for each visit. In drug data analysis, a particular drug, e.g. a psychotropic drug, is considered any-listed if a drug of that class is listed at least once among all of the drugs listed for a visit. For example, whether a visit has one or two antidepressant drugs listed, the visit is counted only once as an “antidepressant visit” and only one of the antidepressant drugs for that visit is counted in summary measures of the number of drugs used by a given population.

DRUG RECODING PROGRAM STEPS

As show in table 1, a drug can be identified by codes from three classification systems: the drug entry code, generic name code and NDC therapeutic drug class. In the example, the generic name code is used in the program for categorizing psychotropic drugs into five basic groups:

- 1 = Anti-anxiety, Hypnotics
- 2 = Non-SSRI Antidepressants
- 3 = SSRI Antidepressants
- 4 = Anti-psychotic Drugs
- 5 = Stimulant Drugs

The steps in the program are:

- Create a format for psychotropic drugs
- Apply the format GENERICF for drug recoding
- Create a drug categorical variable
- Recode drugs from “any-listed” drugs
- Categorize visit data using information from drug category variables

CREATE A FORMAT FOR PSYCHOTROPIC DRUGS

We obtained drug information on psychotropic drug categories by asking experts and using library and Internet resources. We classified psychotropic drugs into five broad categories based on this information. Then, we used VALUE statements to create a numerical format GENERICF with five categories for psychotropic drugs as shown above and two extra categories for other drugs (6 = non-psychotropic drug, and 7 = undetermined drug). For character type drug variables, a character format is needed for the recoding or the variable must be converted into a numerical data type. Format PSYCTYPE is also created for displaying a value of the category number for a numerical variable as a category name used in statistical Report 1. The statements for the format are written as:

```
PROC FORMAT;
*A numerical format that prints generic name code values as drug categorical numbers in character string*;
VALUE GENERICF
50113, 50250, 50713, 50720, 51040, 51055, 51066, 51280, 51695, 52068,
52100, 52110, 52350, 52445, 52503, 53145, 53370, 53515, 53675, 53575,
54085, 54295, 54395, 54760, 55028, 55185, 55668, 55928, 57048 = '1'
50230, 50258, 50714, 92503, 51605, 51955, 52745, 52895, 53248, 59714,
57176, 53975, 54375, 54950, 55900, 55903, 56005, 57096 = '2'
80006, 57064, 57150, 56635, 59829 = '3'
50040, 51150, 51160, 51303, 52335, 52520, 53120, 53150, 53435, 53750,
```

```

54320, 54504, 54850, 57126, 55785, 55795, 55960, 59739, 59644, 54505 = '4'
51665, 53485, 53655, 54240 = '5'
50000 = '7'
. = '8'
OTHER = '6' ;
* A numerical format that prints values of categorical data as drug category names *;
VALUE PSYCTYPE
0 = 'Psychotropic Drug'
1 = 'Anti-anxiety, Hypnotics'
2 = 'Non-SSRI Antidepressants'
3 = 'SSRI Antidepressants'
4 = 'Anti-psychotics Drug'
5 = 'Stimulants Drug'
6 = 'Other Drugs'
7 = 'Undetermined Drug'
8 = 'Missing or blank';
RUN;

```

APPLY THE FORMAT GENERICF FOR DRUG RECODING

First, use the PUT() function to convert each drug code value for the variable DrugCode to a character string specified by the format GENERICF. Then the INPUT() function converts the character value back to a numerical value. The SAS statement for this process is displayed in sample code 1.

Sample Code 1.

```

DrugCat = PUT (DrugCode, GENERICF.);
DrugCategory = INPUT (DrugCat, 2.0);

```

where DrugCategory is a numerical variable with categorical values of 1,2,...n specified in the format GENERICF. An example of drug recoding using sample code 1 is:

DrugCode	50113	50230	55795
DrugCat	'1'	'2'	'4'
DrugCategory	1	2	4

If you are recoding drug data using the generic drug codes directly, the equivalent SAS code for sample code 1 is:

```

IF DrugCode in (50113, 50250, 50713, 50720, 51040, 51055, 51066, 51280,
51695, 52068, 52100, 52110, 52350, 52445, 52503, 53145,
53370, 53515, 53675, 53575, 54085, 54295, 54395, 54760,
55028, 55185, 55668, 55928, 57048) THEN DrugCategory = 1;
ELSE IF DrugCode in (50230, 50258, 50714, 92503, 51605, 51955, 52745,
52895, 53248, 59714, 57176, 53975, 54375, 54950,
55900, 55903, 56005, 57096) THEN DrugCategory = 2;
ELSE IF DrugCode in (80006, 57064, 57150, 56635, 59829) THEN DrugCategory = 3;
ELSE IF DrugCode in (50040, 51150, 51160, 51303, 52335, 52520, 53120,
53150, 53435, 53750, 54320, 54504, 54850, 57126,
55785, 55795, 55960, 59739, 59644, 54505) THEN DrugCategory = 4;
ELSE IF DrugCode in (51665, 53485, 53655, 54240) THEN DrugCategory = 5;
ELSE IF DrugCode in (50000) THEN DrugCategory = 7;
ELSE IF DrugCode in ( . ) THEN DrugCategory = 8;
ELSE DrugCategory = 6;

```

The drug recoding is greatly simplified when sample code 1 is applied, compared to using drug codes directly in SAS IF/THEN statements. The other advantage to using sample code 1 is that you can make changes to the category in PROC FORMAT without changes to the program code. This minimizes potential coding errors. Sample code 1 is only for single ingredient drugs. When drug composition status is considered and combination-ingredient drugs are included, the statements are rewritten as:

Sample Code 2.

```

IF COMPOSIT = 1 THEN DO; *For single entity drug, using the generic name code*;
  GNAMECAT = INPUT (PUT (GENNAME, GENERICF.), 2.0);
END;
IF COMPOSIT = 2 THEN DO; *For combination drug, using the five ingredient code *;
  INGR1 = INPUT (PUT (INGRED1, GENERICF.), 2.0);
  INGR2 = INPUT (PUT (INGRED2, GENERICF.), 2.0);
  INGR3 = INPUT (PUT (INGRED3, GENERICF.), 2.0);

```

```

INGR4 = INPUT (PUT (INGRED4, GENERICF.), 2.0);
INGR5 = INPUT (PUT (INGRED5, GENERICF.), 2.0);
      END;
IF COMPOSIT = 3 THEN DO; *For composite status undetermined drug*;
  GNAMECAT = INPUT (PUT (GENNAME, GENERICF.), 2.0);
      END;

```

where GNAMECAT, INGR1-INGR5 are temporary variables that hold the results of drug recoding. A new drug categorical variable will be created based on the value of those variables in sample code 4.

CREATE A DRUG CATEGORICAL VARIABLE

For a combination drug with several ingredients, if it can only fall into one therapeutic drug category, the hierarchic order of the drug category needs to be set up correctly prior to drug recoding. The hierarchic order of the drug categories can be changed by changing the format specification in the FORMAT procedure.

A macro EXPRESSION in sample code 3 generates an expression that is repeatedly used in an IF/THEN statement in sample code 4 to determine which category the combination drug belongs to based on five drug ingredients.

Sample Code 3.

```

%MACRO EXPRESSION(NUM,CAT,InName);
%DO I = 1 %TO &NUM;
  %IF &I < &NUM %THEN &InName.&I = &CAT OR ;
  %IF &I = &NUM %THEN &InName.&I = &CAT ;
%END;
%MEND EXPRESSION;

```

where argument InName is a literal string that is used to compose variable names with suffix 1 to NUM; NUM is the number of times that the string will be repeated in the expression; CAT is a category specified to be compared with in the expression. When the macro is invoked with NUM = 5, CAT = 1, InName = INGR as show in sample code 4, it generates an expression, INGR1 = 1 OR INGR2 = 1 OR INGR3 = 1 OR INGR4 = 1 OR INGR5 = 1, that evaluates if the drug category of 1 occurs in any of the five ingredient codes.

Variable PSYDRUG is created in sample code 4 to hold the category values from the drug recoding process of sample code 2.

Sample Code 4.

```

IF COMPOSIT = 1 THEN DO; *For single entity drug*;
  PSYDRUG = GENNAME;
      END;
IF COMPOSIT = 2 THEN DO; *For combination drug*;
  IF %EXPRESSION(5, 1, INGR) THEN PSYDRUG = 1;
  ELSE IF %EXPRESSION(5, 2, INGR) THEN PSYDRUG = 2;
  ELSE IF %EXPRESSION(5, 3, INGR) THEN PSYDRUG = 3;
  ELSE IF %EXPRESSION(5, 4, INGR) THEN PSYDRUG = 4;
  ELSE IF %EXPRESSION(5, 5, INGR) THEN PSYDRUG = 5;
  ELSE IF %EXPRESSION(5, 6, INGR) THEN PSYDRUG = 6;
  ELSE IF %EXPRESSION(5, 7, INGR) THEN PSYDRUG = 7;
      END;
IF COMPOSIT = 3 THEN DO; *For composite status undetermined drug*;
  PSYDRUG = GNAMECAT;
      END;

```

Note that the SAS code for drug recoding will become extremely lengthy if you do not use PUT() and INPUT() functions and SAS MACRO language.

RECODE DRUGS FROM "ANY-LISTED" DRUGS

In the 1995-2002 NAMCS data, up to six drugs were recorded. In 2003 and later years, up to eight drugs were recorded. To identify any psychotropic drugs recorded during a visit, one approach is to write the code shown in sample codes 2 and 4 for each drug repeatedly. A more efficient approach is to create flexible SAS code that can be reused for recoding any of the drugs listed for a visit. The drug recoding process in sample codes 2 and 4 are rewritten and combined as a macro PSYCODE with five arguments as shown in sample code 5:

Sample Code 5.

```

%MACRO PSYCODE (COMPOSIT, GENNAME, INGRED, PSYDRUG, FORMAT);
*Convert drug code into drug category *;
IF &COMPOSIT = 1 THEN DO; *For single entity drug*;
  GNAMECAT =INPUT (PUT (&GENNAME, &FORMAT), 2.0);
      END;
ELSE IF &COMPOSIT = 2 THEN DO; *For combination drug*;

```

```

INGR1 = INPUT (PUT (&INGRED.1, &FORMAT), 2.0);
INGR2 = INPUT (PUT (&INGRED.2, &FORMAT), 2.0);
INGR3 = INPUT (PUT (&INGRED.3, &FORMAT), 2.0);
INGR4 = INPUT (PUT (&INGRED.4, &FORMAT), 2.0);
INGR5 = INPUT (PUT (&INGRED.5, &FORMAT), 2.0);
      END;
ELSE IF &COMPOSIT = 3 THEN DO; *For composite status undetermined drug*;
      GNAMECAT = INPUT (PUT (&GENNAME, &FORMAT), 2.0);
      END;
*****;
*Assign recoding results from above processing to the macro variable &PSYDRUG. *
* Note: The coding below can process a format with 10 or less categories. *
* If the number of categories is > 10, the code needs to be revised. *;
IF &COMPOSIT = 1 THEN DO; *For single entity drug*;
      &PSYDRUG = GNAMECAT;
      END;
ELSE IF &COMPOSIT = 2 THEN DO; *For combination drug*;
      IF %EXPRESSION(5, 1, INGR) THEN &PSYDRUG = 1;
      ELSE IF %EXPRESSION(5, 2, INGR) THEN &PSYDRUG = 2;
      ELSE IF %EXPRESSION(5, 3, INGR) THEN &PSYDRUG = 3;
      ELSE IF %EXPRESSION(5, 4, INGR) THEN &PSYDRUG = 4;
      ELSE IF %EXPRESSION(5, 5, INGR) THEN &PSYDRUG = 5;
      ELSE IF %EXPRESSION(5, 6, INGR) THEN &PSYDRUG = 6;
      ELSE IF %EXPRESSION(5, 7, INGR) THEN &PSYDRUG = 7;
      ELSE IF %EXPRESSION(5, 8, INGR) THEN &PSYDRUG = 8;
      ELSE IF %EXPRESSION(5, 9, INGR) THEN &PSYDRUG = 9;
      ELSE IF %EXPRESSION(5, 10, INGR) THEN &PSYDRUG = 10;
      END;
ELSE IF &COMPOSIT = 3 THEN DO; * For composite status undetermined drug*;
      &PSYDRUG = GNAMECAT;
      END;
%MEND PSYCODE;

```

Arguments COMPOSIT, GENNAME, and INGRED pass drug-related information (see Table 1) into the macro PSYCODE for drug recoding. The argument PSYDRUG is a new name for the drug category variable that contains the results from the macro. The argument FORMAT passes a data format for recoding the generic drug code GENNAME. You can use the macro repeatedly for recoding any number of drugs recorded during a visit. Sample macro code for recoding drug 2 in the 2002 NAMCS data file using the macro PSYCODE is

```
%PSYCODE(COMSTAT2, GEN2, DRG2ING, PSYDRUG2, $cGENERICF.);
```

COMSTAT2, GEN2 and DRG2ING contain drug information for drug 2. PSYDRUG2 is the drug category of drug 2. Because GEN2 and DRG2ING1-DRG2ING5 are character variables, a character format \$cGENERICF is necessary for the drug recoding. The macro PSYCODE can be used for recoding any drugs listed in a data set. The macro can categorize each drug into up to ten categories without code modification, which greatly simplifies the coding process. The macro also can be used for categorizing other data into meaningful groups, such as ICD (International Classification of Diseases) diagnosis codes, CPT (Current Procedural Terminology) procedure codes, and other classification systems.

CATEGORIZE VISIT DATA USING INFORMATION FROM DRUG CATEGORY VARIABLES

The purpose of categorizing drug information in the example is to identify the characteristics of visits that are associated with any of the listed drugs. For example, what is the percentage of visits with antidepressant drugs recorded among visits with at least one drug? To answer such questions we need to check all the drug category variables PSYDRUG#, # = 1... 6 for 2002 data, produced by the macro PSYCODE, to find out whether a specific drug category occurs in any of the drug category variables, or whether it occurs at least one time in the six drug category variables. The process of categorizing visit data is written as a macro VISITCAT in sample code 6.

Sample Code 6.

```

%MACRO VISITCAT (DrugNum, InDName, OutDName, CAT);
IF %EXPRESSION(&DrugNum, &CAT, &InDName) THEN &OutDName = 1;
ELSE                                     &OutDName = 2;
%MEND VISITCAT;

```

Argument DrugNum is the number of drugs recorded during a physician office visit in the data set. Argument InDName passes a name into the macro for building an expression using the macro EXPRESSION in sample code 3. Argument OutDName specifies a new variable name for storing the recoding result from the macro VISITCAT.

Argument CAT passes a specific drug category into the macro for checking whether the drug category appears in any of the listed drugs. If the expression is true the new variable is coded as 1; otherwise it is coded as 2. Examples of calling the macro VISITCAT for different psychotropic drug visits are shown in macro DRUGRECODE in sample code 7.

USE DRUG DATA RECODING MACRO IN DATA STEP

Sample code 7 uses FORMAT \$cGENERICF and two macro programs PSYCODE and VISITCAT to recode visit data according to major categories of psychotropic drugs.

Sample Code 7.

```
%MACRO DRUGRECODE(DrugNum, CatNum);
*Drug recoding*;
%DO k = 1 %TO &DrugNum;
    %PSYCODE(COMSTAT&k,GEN&k,DRG&k.ING,PSYDRUG&k,$cGENERICF.);
%END;
*Recode visits with specific psychotropic drugs*;
%DO j = 1 %TO &CatNum;
    %VISITCAT (&DrugNum, PSYDRUG, PSY&J, &J);
%END;
*Recode visits with any psychotropic drugs*;
%VISITCAT (&CatNum, PSY, Psychot, 1);
*Recode visits with any ANTIDEPRESSANT drugs*;
IF PSY2 = 1 OR PSY3=1 THEN Antidep = 1;
ELSE                               Antidep = 2;
%MEND DRUGRECODE;
```

First, the macro PSYCODE is called in a loop to create categorical variables PSYDRUG# for each drug listed; the suffix # denotes numbers from 1 to DrugNum, using drug-related information and the format \$cGENERICF. Then, the macro VISITCAT is invoked five times to create five variables that correspond to five major categories of psychotropic drugs:

```
PSY1 -- Visit with Anti-anxiety, Hypnotics drug
PSY2 -- Visit with Non-SSRI Antidepressants drug
PSY3 -- Visit with SSRI Antidepressants drug
PSY4 -- Visit with Anti-psychotics drug
PSY5 -- Visit with Stimulants drug
```

The five variables can be used directly in data analysis or to create new variables according to the goal of the specific analysis. For example, variable Psychot for visits with any psychotropic drugs is created based on the five categories of psychotropic drugs; variable Antidep for visits with any antidepressant drugs is created based on PSY2 and PSY3 as shown in sample code 7.

The drug-related information specified in sample code 7 is based on the 2002 NAMCS. When calling the macro for recoding data other than the 2002 NAMCS, you need to specify the value of the arguments according to the drug-related information in that data file. Assume the 2002 NAMCS SAS data set **namcs2002** is stored in library 'c:\data'. The invocation of the macro DRUGRECODE for categorizing visits into five major psychotropic drug categories using all six drugs in the 2002 data is:

```
LIBNAME DataDir 'c:\data';
DATA namcs02;
SET DataDir.namcs2002;
%DRUGRECODE(6, 5);
RUN;
```

STATISTICAL REPORT ON CATEGORIZED DRUG DATA

For evaluating the results of categorized drug data, a macro SREPORT produces a two-way tabular report using PROC TABULATE as shown in sample code 8.

Sample Code 8.

```
%MACRO SREPORT(FILEIN, VAR_ROW, VAR_COL );
PROC TABULATE DATA = &FILEIN missing;
WHERE NUMMED>0;                               *Include visits with at least one drug prescribed *;
CLASS &VAR_ROW &VAR_COL;
VAR COUNT;
TABLE &VAR_ROW, (&VAR_COL All)*(N=' N '*F=6.0 PCTN<&VAR_ROW>=' % '*F=6.1
```

```

COUNT='*(SUM=WSUM*F=10.0 PCTSUM<&VAR_ROW>=%*F=6.1))/RTS=30;
TITLE1 "&title" ;
WEIGHT PATWT;
RUN;
%MEND SREPORT;

```

Argument FILEIN specifies the drug data used by PROC TABULATE for the statistical report. Arguments VAR_ROW and VAR_COL specify class variables that determine the categories that PROC TABULATE uses to calculate statistics in each table cell and defines the rows and columns of the table. NUMMED > 0 in the WHERE statement limits the statistical report only to visits with at least one drug recorded during a physician office visit. The analysis variable COUNT in the VAR statement is necessary for computing the weighted statistics, e.g., the weighted total SUM and weighted percentage PCTSUM based on the row variable VAR_ROW in the TABLE statement. Using the macro together with ODS can produce a statistical report in HTML or RTF file format that is easy to import into MS Word documents. Example reports produced by the macro are:

- Percentage of visits with selected major categories of psychotropic drugs recorded as the first-listed drug, among visits with at least one drug recorded during physician office visits, by sex.
- Percentage of visits with any psychotropic drugs prescribed among visits with at least one drug recorded during physician office visits, by sex.

SAS CODE FOR REPORT 1:

```

ODS HTML BODY="c:\sas\Drug1report.html"; *specify a ODS HTML file destination*;
%let title=Report 1: Percent of visits with selected major categories of psychotropic drugs recorded as the first-listed
Drug, among visits with at least one drug recorded, by sex: United States, 2002;
%SREPORT(namcs02,psydrug1,sex);
ODS HTML CLOSE;

```

In the code for Report 1, the categorized variable PSYDRUG1 for the first-listed drug is specified as the row variable, and the demographic variable SEX is used as the column variable. The report shows the difference in estimates of visits with selected major psychotropic drugs between males and females. The output is in Report 1.

Report 1: Percent of visits with selected major categories of psychotropic drugs recorded as the first-listed drug, among visits with at least one drug recorded, by sex: United States, 2002

	SEX								All			
	FEMALE				MALE							
	N	%	WSUM	%	N	%	WSUM	%	N	%	WSUM	%
PSYDRUG1												
Anti-anxiety, Hypnotics	165	1.6	4497549	1.3	105	1.4	2841859	1.2	270	1.5	7339408	1.3
Non-SSRI Antidepressants	266	2.6	5692075	1.6	158	2.1	3533828	1.5	424	2.4	9225903	1.6
SSRI Antidepressants	457	4.5	12266894	3.5	251	3.4	6320818	2.7	708	4.0	18587712	3.2
Anti-psychotics Drug	109	1.1	1593255	0.5	110	1.5	1686556	0.7	219	1.2	3279811	0.6
Stimulants Drug	37	0.4	693471	0.2	94	1.3	2388342	1.0	131	0.7	3081813	0.5
Other Drugs	8750	85.4	305726624	88.2	6436	86.1	202891949	88.0	15186	85.7	508618573	88.1
Undetermined Drug	458	4.5	16121380	4.7	317	4.2	10787480	4.7	775	4.4	26908860	4.7
Missing or blank	1	0.0	9684	0.0	1	0.0	23137	0.0	2	0.0	32821	0.0

SAS CODE FOR REPORT 2:

```

ODS HTML BODY="c:\sas\Drugreport2.html";
%LET title=Report 2: Percent of visits with any psychotropic drugs prescribed among visits with at least one drug
recorded during physician office visits, by sex: United States, 2002;
%REPORT(namcs02, psycho, sex);
ODS HTML CLOSE;

```

In the code for Report 2, variable PSYCHO is specified as the row variable, SEX is still the column variable. The report shows the differences in estimates of the percentage of visits with any psychotropic drugs between males and females. The output is in Report 2.

Report 2: Percent of visits with any psychotropic drugs recorded among visits with at least one drug recorded during physician office visits, by sex: United States, 2002

	SEX								All			
	FEMALE				MALE							
	N	%	WSUM	%	N	%	WSUM	%	N	%	WSUM	%
Psychot												
Psychotropic drugs	1796	17.5	49130639	14.2	1150	15.4	28651583	12.4	2946	16.6	77782222	13.5
Other drugs	8447	82.5	297470293	85.8	6322	84.6	201822386	87.6	14769	83.4	499292679	86.5

Comparing Report 1 with Report 2 we find that the estimated percentage of visits with psychotropic drugs prescribed when using the first-listed drug is much less than when using any-listed drugs (7.2% vs. 13.5% using weighted data and 9.8% vs. 16.6% using unweighted data). This shows that you can obtain better estimates of the use of a particular drug if you use any-listed drugs rather than only the first-listed drug.

CONCLUSION

The SAS INPUT(), PUT() functions and SAS MACRO are important tools in SAS programming tasks. A SAS program for categorizing drugs into meaningful groups based on survey (or other) data can be greatly simplified by using the SAS INPUT() and PUT() functions and Macro Language. Using PROC FORMAT, INPUT() and PUT() functions can make your SAS programming easier and neater, and minimize potential coding errors. Using the SAS MACRO program can reduce redundant codes, and increase code reusability.

REFERENCES

1. National Center for Health Statistics. Health, United States, 2004 With Chartbook on Trends in the Health of Americans. Hyattsville, Maryland: 2004.
2. National Center for Health Statistics. Public Use Data File Documentation, 2002 National Ambulatory Medical Care Survey.
3. SAS Institute Inc., SAS Guide to Macro Processing, Version 6, Second Edition, Cary, NC: SAS Institute Inc., 1990.
4. SAS Institute Inc., SAS Procedures Guide, Version 6, Third Edition, Cary, NC: SAS Institute Inc., 1990.
5. SAS Institute Inc., SAS Language: Reference, Version 6, First Edition, Cary, NC: SAS Institute Inc., 1990.

ACKNOWLEDGMENTS

I would like to thank Amy Bernstein Sc.D. for her valuable comments and corrections on the draft of this paper and Patricia Pastor Ph.D. for her specialized knowledge about psychotropic drugs.

CONTACT INFORMATION

Henry Xia
 CDC/National Center for Health Statistics
 3311 Toledo Road, Room 6227
 Hyattsville, MD 20782
 Work Phone: 301-458-4596
 Fax: 301-458-4036
 E-mail: hxia@cdc.gov

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their respective companies.